| # =========================================================================== | |
| # https://www.gnu.org/software/autoconf-archive/ax_append_flag.html | |
| # =========================================================================== | |
| # | |
| # SYNOPSIS | |
| # | |
| # AX_APPEND_FLAG(FLAG, [FLAGS-VARIABLE]) | |
| # | |
| # DESCRIPTION | |
| # | |
| # FLAG is appended to the FLAGS-VARIABLE shell variable, with a space | |
| # added in between. | |
| # | |
| # If FLAGS-VARIABLE is not specified, the current language's flags (e.g. | |
| # CFLAGS) is used. FLAGS-VARIABLE is not changed if it already contains | |
| # FLAG. If FLAGS-VARIABLE is unset in the shell, it is set to exactly | |
| # FLAG. | |
| # | |
| # NOTE: Implementation based on AX_CFLAGS_GCC_OPTION. | |
| # | |
| # LICENSE | |
| # | |
| # Copyright (c) 2008 Guido U. Draheim <[email protected]> | |
| # Copyright (c) 2011 Maarten Bosmans <[email protected]> | |
| # | |
| # Copying and distribution of this file, with or without modification, are | |
| # permitted in any medium without royalty provided the copyright notice | |
| # and this notice are preserved. This file is offered as-is, without any | |
| # warranty. | |
| #serial 8 | |
| AC_DEFUN([AX_APPEND_FLAG], | |
| [dnl | |
| AC_PREREQ(2.64)dnl for _AC_LANG_PREFIX and AS_VAR_SET_IF | |
| AS_VAR_PUSHDEF([FLAGS], [m4_default($2,_AC_LANG_PREFIX[FLAGS])]) | |
| AS_VAR_SET_IF(FLAGS,[ | |
| AS_CASE([" AS_VAR_GET(FLAGS) "], | |
| [*" $1 "*], [AC_RUN_LOG([: FLAGS already contains $1])], | |
| [ | |
| AS_VAR_APPEND(FLAGS,[" $1"]) | |
| AC_RUN_LOG([: FLAGS="$FLAGS"]) | |
| ]) | |
| ], | |
| [ | |
| AS_VAR_SET(FLAGS,[$1]) | |
| AC_RUN_LOG([: FLAGS="$FLAGS"]) | |
| ]) | |
| AS_VAR_POPDEF([FLAGS])dnl | |
| ])dnl AX_APPEND_FLAG | |